数値計算法

第八回
数値積分法の基礎

若狭　智嗣

粒子物理学講座

数値積分法1－区分求積法－

- **区分求積法の考え方**
  - 積分＝面積を求める
  - 積分領域を N 個の領域に分割
  - 各領域は長方形で近似



$$h = \frac{b-a}{N}$$

- **区分求積法の公式**

$$I = \int_a^b f(x)dx = \sum_{i=1}^{N} f(a+ih)h \qquad h = \frac{b-a}{N}$$

区分求積法の精度

- N個の長方形のうち最初の長方形を考える
  - 区分求積法の積分値　$S = f(a+h)h$
  - $x = a$ の周りのTayler展開で置き換えて

$$S = f(a+h)h = \left[f(a) + hf'(a) + O(h^2) \right] h$$
$$= hf(a) + h^2 f'(a) + O(h^3)$$

- $f(x)$ の不定積分を $F(x)$ とおく（ $F'(x) = f(x)$ ）
  - 真の積分値　$I = \int_a^{a+h} f(x)dx = F(a+h) - F(a)$
  - $x = a$ の周りのTayler展開で置き換えて

$$I = F(a+h) - F(a) = hF'(a) + \frac{h^2}{2} F''(a) + O(h^3)$$
$$= hf(a) + \frac{h^2}{2} f'(a) + O(h^3)$$

$\frac{h^2}{2} f'(a)$ の差

区分求積法の判定法

- **区分求積法の誤差**
  - 各区分に関して $$S - I \approx \frac{h^2}{2} f'(a)$$
  - 全区間(N個)に関して $$\frac{h^2}{2} f' \times \underbrace{\frac{b-a}{h}}_{N} = \frac{h}{2}(b-a) f'$$

- **判定法**
  - 刻み数 N を倍にする(h を半分にする)→誤差は約1/2
  - (正解が分からない)実際には？

$I_N - I_{N/2}$ を計算して、Nを倍にした時に約1/2になる事を確認

演習

- 次の定積分

$$I=\int_1^2 \frac{1}{x^2}dx=\frac{1}{2}$$

  を区分求積法により求めよ。
- 分割数Nを10，100，1000と順次増やして、厳密解との差がどうなるか確認せよ。
- 区分求積法が正しく機能する最大のNはどの程度か？

プログラム(kubun)の説明
(/home/teacher/z6wt01in/SAMPLE/kubun.f)

$$h = \frac{2-1}{N} = \frac{1}{N}$$

$$I = h\sum_{j=1}^{N} f(a+jh) = h\sum_{j=1}^{N}\frac{1}{(1+jh)^2}$$

```
do i=1,7
    n=10**i
    h=1.0e0/real(n)
    s=0.0e0
    do j=1,n
       s = s + 1.0e00/(1.0e+00+real(j)*h)**2
    end do
    write(*,'(i8,2f11.7)') n,s*h,s*h-5.0e-1
end do
```

分割数が10^1から10^7の場合を計算

分割数、数値計算の値、厳密解との差

区分数依存性

- 結果

	区分数	計算解	厳密解との差
-	10	0.4639551	-0.0360449
-	100	0.4962645	-0.0037355
-	1000	0.4996252	-0.0003748
-	10000	0.4999619	-0.0000381
-	100000	0.4999943	-0.0000057
-	1000000	0.4997711	-0.0002289
-	10000000	0.5491630	0.0491630

Nを10倍にしても、
差が1/10になっていない
（誤差の蓄積のため）

数値微分の場合と同様に、数値積分の場合もむやみと分割数を大きくしても精度は向上しない
（誤差の累積によりかえって精度が悪くなる場合がある）

演習　その2

- 次の定積分

$$I = \int_1^2 \frac{1}{x^2} dx = \frac{1}{2}$$

を倍精度の区分求積法により求めよ。

- 結果を単精度の場合と比較せよ

```
real x   ! x は単精度の実数
real*8 y ! y は倍精度の実数
```

演習　その3

- 次の定積分

$$I=\int_{-1}^{1}\frac{2}{1+x^2}dx=\pi=3.14159265\cdots$$

を単精度と倍精度の区分求積法により求めよ。

- 厳密解にもっとも近い値となる分割数 N は、単精度の場合と倍精度の場合で各々いくらか？

```
real x   ! x は単精度の実数
real*8 y ! y は倍精度の実数
```

数値積分法2－台形法－

- **台形法の考え方**
  - 積分＝面積を求める
  - 積分領域を N 個の領域に分割
  - 各領域は台形で近似

区分求積法　　台形法



- **台形法の公式**

$$I=\int_a^b f(x)dx \approx \sum_{i=0}^{N-1}\frac{f(a+ih)+f(a+(i+1)h)}{2}h$$

N では無いことに注意!!

$$=\frac{f(a)+f(b)}{2}h+\sum_{i=1}^{N-1}f(a+ih)h \qquad h=\frac{b-a}{N}$$

台形法の精度

- N個の台形のうち最初の長方形を考える
  - 台形法の積分値 $S=\frac{h}{2}\left[f(a)+f(a+h)\right]$
  - x=aの周りのTayler展開で置き換えて

$$S=\frac{h}{2}\left[f(a)+f(a+h)\right]=\frac{h}{2}\left[f(a)+f(a)+hf'(a)+\frac{h^2}{2}f''(a)+O(h^3)\right]$$
$$=hf(a)+\frac{h^2}{2}f'(a)+\frac{h^3}{4}f''(a)+O(h^4)$$

$\frac{h^3}{12}f''(a)$
の差

- $f(x)$の不定積分を$F(x)$とおく（$F'(x)=f(x)$）
  - 真の積分値 $I=\int_a^{a+h}f(x)dx=F(a+h)-F(a)$
  - x=aの周りのTayler展開で置き換えて

$$I=F(a+h)-F(a)=hF'(a)+\frac{h^2}{2}F''(a)+\frac{h^3}{6}F'''(a)+O(h^4)$$
$$=hf(a)+\frac{h^2}{2}f'(a)+\frac{h^3}{6}f''(a)+O(h^4)$$

台形法の判定法

- **台形法の誤差**
  - 各区分に関して $$S - I \approx \frac{h^3}{12} f''(a)$$
  - 全区間(N個)に関して $$\frac{h^3}{12} f'' \times \underbrace{\frac{b-a}{h}}_{N} = \frac{h^2}{12}(b-a) f''$$
- **判定法**
  - 刻み数 N を倍にする(h を半分にする)→誤差は約1/4
  - (正解が分からない)実際には？

$I_N - I_{N/2}$ を計算して、Nを倍にした時に約1/4になる事を確認

数値積分法3－Simpson法－

- 台形法の精度

$$\left.\begin{aligned} S_N - I &\approx \frac{h^2}{12}(b-a)f'' \\ S_{N/2} - I &\approx \frac{(2h)^2}{12}(b-a)f'' \end{aligned}\right\} \quad (S_N - I) - \frac{1}{4}(S_{N/2} - I) \approx 0$$

$\Downarrow$ より高精度の式

$$\boxed{I \approx \frac{S_N - S_{N/2}/4}{1-1/4}}$$

Simpson法

- Simpson法の公式

$$I = \int_a^b f(x)dx \qquad h = \frac{b-a}{N}$$

$$\approx \frac{h}{3}\left[f(a) + f(b) + 4\sum_{i=1}^{N/2} f(a+(2i-1)h) + 2\sum_{i=1}^{N/2-1} f(a+2ih)\right]$$

各数値積分法のまとめ

- **積分区間を分割して、各区間の面積を求めるという点では同じ**
- **区分求積法**
  - ある1点を通る定数で被積分関数を近似
  - 各区間は長方形
  - 誤差は h 程度
- **台形法**
  - ある2点を通る直線で被積分関数を近似
  - 各区間は台形
  - 誤差は h^2 程度
- **Simpson法**
  - ある3点を通る2次曲線で被積分関数を近似
  - 各区間は山（谷）形
  - 誤差は h^4 程度

演習

- **積分**

$$I=\int_0^1 e^x dx = e-1 = 1.7182818\cdots$$

を考える

  - 台形法およびSimpson法により数値積分を行え
  - 刻み数Nを変えて計算し、以下の事を確認せよ
    - 収束の度合い（厳密解との差）→ Gnuplotを用いてグラフを描け
    - 各方法が正常に機能しているか

- **積分**

$$I=\int_{-1}^1 \frac{2}{1+x^2} dx = \pi = 3.14159265\cdots$$

に対して、上記と同じことを行え

プログラム(integral)の説明

(/home/teacher/z6wt01in/SAMPLE/integral.f)

```
c begin:
      exact = exp(1.0)-1
      do m=0,6
         n=2**m                          ! 刻み数
         h=1.0/n                         ! 刻み幅
c 台形法
         integral=y(0.0)+y(1.0)    ! 積分端での値
         do i=1,n-1
            x = h*i
            integral=integral+2.0*y(x)
         end do
         trapezoid=integral*h/2.0 ! 台形法の解
```

厳密解

$e-1$

分割数が

2^0から2^6

の場合を計算

台形法の公式 $S_N=\frac{h}{2}\left[f(a)+f(b)+\sum_{i=1}^{N-1}2f(a+ih)\right]$

プログラム(integral)の説明(続き)
(/home/teacher/z6wt01in/SAMPLE/integral.f)

```
c Simpson 法
      integral=y(0.0)+y(1.0)    ! 積分端での値
      do i=1,n/2
         x = h*(2*i-1)
         integral=integral+4.0*y(x)
      end do
      do i=1,n/2-1
         x = h*(2*i)
         integral=integral+2.0*y(x)
      end do
      simpson=integral*h/3.0    ! Simpson 法の解
c 数値積分の結果と解析解の差の出力
```

Simpson法の公式 $$S_N = \frac{h}{3}\left[f(a) + f(b) + 4\sum_{i=1}^{N/2} f(a+(2i-1)h) + 2\sum_{i=1}^{N/2-1} f(a+2ih) \right]$$

プログラム(integral)の説明(続き)
(/home/teacher/z6wt01in/SAMPLE/integral.f)

```
c
c 被積分関数の定義
c
      real function y(x)
      implicit none
c input:
      real  x
c begin:
      y = exp(x)
      return
      end
```

被積分関数を単精度実数の関数で定義

$$I = \int_0^1 e^x dx$$

結果の分割数依存性

- **予想**
  - 台形法　　分割数を倍(hを1/2)にすれば厳密解との差(誤差)は1/4
  - Simpson法 分割数を倍(hを1/2)にすれば厳密解との差(誤差)は1/16
- **結果**

N	h	台形法の値と差		Simpson法の値と差	
1	1.00000	1.8591409	0.1408591	1.2394272	−0.4788545
2	0.50000	1.7539310	0.0356493	1.7188611	0.0005794
4	0.25000	1.7272218	0.0089401	1.7183189	0.0000372
8	0.12500	1.7205185	0.0022367	1.7182842	0.0000025
16	0.06250	1.7188411	0.0005593	1.7182817	0.0000000
32	0.03125	1.7184218	0.0001401	1.7182817	0.0000000
64	0.01562	1.7183169	0.0000352	1.7182817	0.0000000

前の値の1/4

前の値の1/16

gnuplotを使った視覚化

- **結果をファイルに書き出す**
  - % integral > integral.dat
- **gnuplotを起動して、絵を描かせる**
  - plot 'integral.dat' using 1:3 with linespoints 1 5,'integral.dat' using 1:5 with linespoints 3 5



赤が台形法
青がSimpson法

不連続点がある場合の数値積分

- **物理学に現れる被積分関数**
  - 発散点や不連続点がある場合がある
  - 発散点があっても、積分はある有限な物理的に意味のある場合がある
    → 発散点や不連続点のある関数の精度のよい数値積分は重要

- **発散点や不連続点のある場合の処方箋**
  - 被積分関数 ***f(x)*** の不連続点や、その高階の導関数が不連続になる点が予め知られている場合は、その点を ***c*** として、積分区間を、

$$\int_a^b f(x)dx = \int_a^c f(x)dx + \int_c^b f(x)dx$$

と分割すればよい

不連続点がある場合の数値積分の例

- **例題**

2階導関数が
x=1/3で不連続

$$f(x) = \begin{cases} 1 & 0 \le x \le \frac{1}{3} \\ 1 - \frac{9}{4}\left(x - \frac{1}{3}\right)^2 & \frac{1}{3} \le x \le 1 \end{cases}$$

$$I = \int_0^1 f(x)dx = \frac{7}{9}$$

- **積分区間を分割しない場合**

N	台形法 差	Simpson法 差
2	−0.0590278	0.0138889
4	−0.0160590	−0.0017361
8	−0.0038520	0.0002170
16	−0.0009834	−0.0000271
32	−0.0002433	0.0000034

前の値の
約1/4
→正常

前の値の
約1/8
→異常

不連続点がある場合の数値積分の例

- **例題**

2階導関数が
x=1/3で不連続

$$f(x) = \begin{cases} 1 & 0 \le x \le \frac{1}{3} \\ 1 - \frac{9}{4}\left(x - \frac{1}{3}\right)^2 & \frac{1}{3} \le x \le 1 \end{cases} \rightarrow \boxed{I = \int_0^1 f(x)dx = \frac{7}{9}}$$

- **積分区間を分割した場合としない場合の比較**
（分割した場合は、[0,1/3]と[1/3,1]を各々(N−1)で分割）

	N	台形法の差		Simpson法の差	
		分割なし	分割あり	分割なし	分割あり
-	2	−0. 0590278	−0. 1111111	0. 0138889	−0. 3333333
-	4	−0. 0160590	−0. 0277777	−0. 0017361	0. 0000000
-	8	−0. 0038520	−0. 0069444	0. 0002170	0. 0000000
-	16	−0. 0009834	−0. 0017362	−0. 0000271	0. 0000000
-	32	−0. 0002433	−0. 0004340	0. 0000034	0. 0000000

収束

前の値の約1/4
→正常

演習（不連続点のある場合の数値積分）

- 次の積分

$$f(x)=\begin{cases}1 & 0\le x\le \frac{1}{3}\\ 1-\frac{9}{4}\left(x-\frac{1}{3}\right)^2 & \frac{1}{3}\le x\le 1\end{cases}$$

を台形法およびSimpson法により数値積分を行え。

- 不連続点x=1/3で積分区間を分割した場合としない場合両方に対して計算を行い、講義ノート94ページの表１０．１を確認せよ。

プログラム例（被積分関数の表し方）

```
      real function y(x)
      implicit none
c input:
      real x
c begin:
      if (x.lt.1.0/3.0) then
         y = 1.0
      else
         y = 1.0-9.0/4.0*(x-1.0/3.0)**2
      endif
      return
      end
```

$$f(x) = \begin{cases} 1 & 0 \leq x \leq \frac{1}{3} \\ 1 - \frac{9}{4}\left(x - \frac{1}{3}\right)^2 & \frac{1}{3} \leq x \leq 1 \end{cases}$$

if文を使って
定義域毎に
関数を定義する

プログラム例

```
c
      real A,B
      parameter(A=0.0,B=1.0)          ! 積分範囲 [a,b]
      real C                          ! 分割点
      parameter(C=1.0/3.0)
```

途中省略

```
      do j=1,5
        n  = 2**j
        h  = (B-A)/n
        n1 = 2**(j-1)
        h1 = (C-A)/n1
        h2 = (B-C)/n1
c 台形法（分割なし）
        sum1 = 0.0
        do i=1,n-1
          sum1 =sum1 + y(A+i*h)
        end do
        trapezoid = h/2.0*(y(A)+y(B)+2.0*sum1)
```

[A,B]を $n = 2^j$ で分割

[A,C]と[C,B]を $n1 = 2^{j-1}$ で分割

→[A,B]は$n = 2n1=2^j$ で分割

プログラム例（つづき）

```
c 台形法（分割あり）
      sum1 = 0.0
      sum2 = 0.0
      do i=1,n1-1
         sum1 =sum1 + y(A+i*h1)
         sum2 =sum2 + y(C+i*h2)
      end do
      trapezoid2 = h1/2.0*(y(A)+y(C)+2.0*sum1)
     &             +h2/2.0*(y(C)+y(B)+2.0*sum2)
```

区間[A,C]に対して
台形法で計算

区間[C,B]に対して
台形法で計算